Dot Products and Angles via *Mathematica*

Department of Mathematics, 196 Auditorium Road Unit 3009, Storrs CT 06269-3009.

One of the fundamental characteristics of vectors is *direction*, so it is important to have a means of measuring the angle between two vectors. As a first step, consider the angle α between the positive x-axis and the vector $\mathbf{v} = (a, b) = a\mathbf{i} + b\mathbf{j}$. From the figure below, its cosine is $a/\|\mathbf{v}\|$. That



is the quotient of the *x*-coordinate of the endpoint of **v** by the length of **v**:

(1) $$\cos\alpha = \frac{a}{\sqrt{a^2+b^2}} = \frac{a\cdot 1 + b\cdot 0}{\|\mathbf{v}\|\ \|\mathbf{i}\|}.$$

The last expression emphasizes that cos α is the sum of the products of the corresponding coordinates of **v** = (*a*, *b*) and **i** = (1, 0) divided by the product of the lengths of **v** and **i**.

Similarly, the cosine of the angle β between **v** and the positive *y*-axis is

(2) $$\cos\beta = \frac{b}{\sqrt{a^2+b^2}} = \frac{a\cdot 0 + b\cdot 1}{\|\mathbf{v}\|\ \|\mathbf{j}\|},$$

where notice that once again the formula is the sum of the products of the respective coordinates of the two vectors **v** and **j** divided by the product of their lengths.

Next, consider the more general situation of two nonzero vectors **v** = (*a*, *b*) and **w** = (*c*, *d*) and the angle θ between them. As the next figure shows, that angle is the difference between the angles γ and α that the two vectors make with the positive *x*-axis.



Thus,

$$\cos\theta = \cos(\gamma - \alpha) = \cos\gamma\ \cos\alpha + \sin\gamma\ \sin\alpha$$
$$= \frac{c\,a}{\|\mathbf{v}\|\,\|\mathbf{w}\|} + \frac{d\,b}{\|\mathbf{v}\|\,\|\mathbf{w}\|} = \frac{a\,c + b\,d}{\|\mathbf{v}\|\,\|\mathbf{w}\|}.$$

The last expression has the same form as those in (1) and (2): the numerator is the sum of the products of the respective coordinates of the two vectors, this time **v** and **w**. The denominator is the product of the lengths of those vectors. The numerator is a very important quantity.

2**.1. Definition.** If **v** = (*a*, *b*) and **w** = (*c*, *d*) are two vectors in the plane, then their ***dot***

product is

(3) $$\mathbf{v} \cdot \mathbf{w} = ac + bd.$$

Similarly, if $\mathbf{v} = (a, b, c)$ and $\mathbf{w} = (p, q, r)$ are two vectors in 3-space, then their dot product is the sum of the products of the corresponding coordinates:

(4) $$\mathbf{v} \cdot \mathbf{w} = ap + bq + cr.$$

In general, if $\mathbf{v} = (v_1, v_2, \ldots, v_n)$ and $\mathbf{w} = (w_1, w_2, \ldots, w_n)$ are n-dimensional vectors, then their dot product is the sum of the products of the corresponding coordinates:

(5) $$\mathbf{v} \cdot \mathbf{w} = v_1 w_1 + v_2 w_2 + \ldots + v_n w_n.$$

With this notation, the above formula for the cosine of the angle between two vectors in the plane becomes

(6) $$\cos\theta = \frac{\boldsymbol{v} \cdot \boldsymbol{w}}{\|\boldsymbol{v}\| \, \|\boldsymbol{w}\|}.$$

The dot product has many of the familiar algebraic properties of multiplication of real numbers. The following theorem from Section 1.2 lists several. Here the vectors can be of any dimension.

2.4 Theorem. For all vectors $\mathbf{x}$, $\mathbf{y}$, and $\mathbf{z}$ and any real number a,

(a) $\mathbf{x} \cdot \mathbf{y} = \mathbf{y} \cdot \mathbf{x}$

(b) $\mathbf{x} \cdot (\mathbf{y} + \mathbf{z}) = \mathbf{x} \cdot \mathbf{y} + \mathbf{x} \cdot \mathbf{z}$ and $\mathbf{x} \cdot (a\mathbf{y}) = a\,(\mathbf{x} \cdot \mathbf{y})$

(c) $\mathbf{0} \cdot \mathbf{x} = 0$ for all $\mathbf{x}$, and if $\mathbf{v} \cdot \mathbf{x} = \mathbf{0}$ for all $\mathbf{x}$, then $\mathbf{v} = \mathbf{0}$.

(d) $\mathbf{x} \cdot \mathbf{x} \geq 0$ for all $\mathbf{x}$, and $\mathbf{x} \cdot \mathbf{x} = 0$ only if $\mathbf{x} = \mathbf{0}$

(e) $\mathbf{x} \cdot \mathbf{x} = \|\mathbf{x}\|^2$

Proof. Parts (a) and (b) are an easy verification, which you can supply.

(c) $\mathbf{0} \cdot \mathbf{x} = 0$ follows directly from the definition of dot product. Suppose next that $\mathbf{v} \cdot \mathbf{x} = \mathbf{0}$ for all $\mathbf{x}$, where $\mathbf{v} = (v_1, v_2, \ldots, v_n)$. Then in particular for $\mathbf{x} = \mathbf{e}_i$, where $\mathbf{e}_i$

is the *i*th standard basis vector, it is true that $\mathbf{v} \cdot \mathbf{e}_i = 0$. But since $\mathbf{e}_i = (0, 0, ..., 0, 1, 0, ..., 0)$, where the 1 occurs in coordinate place *i*, it follows that for any value of *i*,

$$\mathbf{v} \cdot \mathbf{e}_i = v_i = 0.$$

Thus, $\mathbf{v} = \mathbf{0}$.

(d), (e) For any $\mathbf{x} = (x_1, x_2, ..., x_n)$, note first that

$$\mathbf{x} \cdot \mathbf{x} = (x_1)^2 + (x_2)^2 + ... + (x_n)^2 = \| \mathbf{x} \|^2 \geq 0.$$

This establishes (e) and half of (d). Since the only way for a sum of squares to be 0 is for each summand to be 0, it follows that $\mathbf{x} \cdot \mathbf{x} = 0$ only in case each $x_i = 0$, that is, $\mathbf{x} = \mathbf{0}$. QED

The above properties have special names: (a) is the *symmetric* property, (b) the *bilinear* property, (c) and (d) the *nondegeneracy* property of the dot product. (These concepts are important in linear algebra, but do not play a major role in vector caculus.)

Hand calculation of dot products involves only simple multiplication and addition. For example, consider the dot product of the vectors **v** = (–1, 2, 3) and **w** = (3, –1, 2) in 3-space and the dot product of the plane vectors **v** = (1, 2) and **w** = (3, 1). *Mathematica* has a built-in command `Dot` for calculating dot products, and you can use it to check your arithmetic if you like — although accessing it may be more trouble than the benefits justify. Here is a simple call to `Dot`, which you can execute as usual by moving your cursor to the end of the last line, and hitting the Enter key.

```
v := {-1, 2, 3};
w := {3, -1, 2};
v.w
```

```
1
```

A more interesting use of *Mathematica* is to calculate the angle in radians between two vectors **v** and **w** from (6). To do that, invoke *Mathematica*'s built-in `ArcCos` command:

```
ArcCos[ v.w / Sqrt[(v.v) (w.w)] ]
```

$$\text{ArcCos}\left[\frac{\mathtt{v.w}}{\sqrt{(\mathtt{v.v})\,(\mathtt{w.w})}}\right]$$

$$\text{ArcCos}\left[\frac{1}{14}\right]$$

Since *Mathematica* sees nothing but integers in its input from the two vectors, and since the expression on the right side of (6) turns out to be rational, it outputs the symbolic expression for the inverse cosine of that rational number. To get a numerical approximation of the number of radians that output represents, invoke the numerical approximation function `N[ ]`. Try it!

```
N[ArcCos[v.w / √((v.v) (w.w))]]
```

```
1.49931
```

Mathematica can draw a labeled picture of the two vectors **v** and **w** and report the angle between them. The following routine does so for two-dimensional vectors. Execute it to see the result.

```
Needs["Graphics`Arrow`"]
v := {1, 2};
w := {3, 1};
Show[Graphics[{{RGBColor[1, 0, 0], Arrow[{0, 0}, v],
    Text[FontForm["v", {"Times-Bold", 12}], {.9 v[[1]], .8 v[[2]]}]},
   {RGBColor[1, 0, 1], Text[FontForm["θ", {"Symbol", 12}],
     {0.2 v[[1]], 0.1 v[[2]]}], Arrow[{0, 0}, w],
    Text[FontForm["w", {"Times-Bold", 12}], {.9 w[[1]], .8 w[[2]]}]}},
  Axes → True, AxesLabel → {"x", "y"}]]
```



Mathematica's lack of a 3-dimensional Arrow command means that to produce a similar 3-dimensional picture is more challenging. The following long program handles a pair of three-dimensional vectors. Execute it to view the picture.

```
v = {1, 2, 3};
w = {-1, 1, 1};
q = √v.v;
r = √w.w;
xwin = 2;
```

```
yext = 2;
zlen = 4;
coordaxes = Graphics3D[
  {{RGBColor[0, 1, 0], Line[{{-xwin, 0, 0}, {xwin, 0, 0}}],
    Text["x", {xwin + xwin/5, 0, 0}]},
   {RGBColor[0, 1, 0], Line[{{0, -yext, 0}, {0, yext, 0}}],
    Text["y", {0, yext + yext/5, 0}]},
   {RGBColor[0, 1, 0], Line[{{0, 0, -zlen}, {0, 0, zlen}}],
    Text["z", {0, 0, zlen + zlen/5}]}}];
vplot = Graphics3D[{{RGBColor[1, 0, 0], Line[{{0, 0, 0}, v}],
     Polygon[{v, {.9 v[[1]] - yext/4, .9 v[[2]] + xwin/4, .9 v[[3]] - 1/q},
       {.9 v[[1]] + yext/4, .9 v[[2]] - xwin/4, .9 v[[3]] + 1/q}}],
     Text[FontForm["v", {"Times-Bold", 12}],
      {.5 v[[1]], .5 v[[2]], .5 v[[3]] + 2/q}]},
    {RGBColor[1, 0, 1], Line[{{0, 0, 0}, w}],
     Polygon[{w, {.9 w[[1]] - yext/4, .9 w[[2]] + xwin/4, .9 w[[3]] - 1/r},
       {.9 w[[1]] + yext/4, .9 w[[2]] - xwin/4, .9 w[[3]] + 1/r}}],
     Text[FontForm["w", {"Times-Bold", 12}],
      {.5 w[[1]], .5 w[[2]], .5 w[[3]] + 1/r}]},
    Text[FontForm["θ", {"Symbol", 12}],
     {0.1 v[[1]], 0.1 v[[2]], 0.2 v[[3]] + 1/q}]}];
Show[vplot, coordaxes, Axes → True, Lighting → False]
```



The dot product has a number of important properties, among the most useful of which is the next one.

2.6. Theorem. Cauchy-Schwarz Inequality. For any two vectors **x** and **y** in $\mathbf{R}^n$,

(7) $$| \mathbf{x} \cdot \mathbf{y} | \leq \| \mathbf{x} \| \, \| \mathbf{y} \|.$$

For a proof, refer to your text. Note that (7) illustrates the advantage of the notation **||v||** for the length of a vector **v**. It makes clear that the Cauchy-Schwarz inequality says that the *size* (absolute value) of the real number **x · y** is no greater than the product of the *magnitudes* (lengths) of the two vectors. A simple illustration of (7) comes from the vectors above: **v** = (1, 2, 3) and **w** = (–1, 1, 1). Since

$$\mathbf{v} \cdot \mathbf{w} = -1 + 2 + 3 = 4, \;\; \| \mathbf{v} \| = \sqrt{1 + 4 + 9}, \;\text{ and }\; \| \mathbf{w} \| = \sqrt{1+1+1},$$

the inequality indeed holds for these two vectors:

$$| \mathbf{v} \cdot \mathbf{w} | = 4 \leq \sqrt{14} \sqrt{3} = \sqrt{42.}$$

The importance of this inequality is far greater than this calculation suggests. It permits definition of the angle between vectors in n-dimensional space for $n > 3$. Let **x** and **y** be nonzero vectors of any dimension, and rewrite (7) as follows.

$$-\| \mathbf{x} \| \, \| \mathbf{y} \| \leq \mathbf{x} \cdot \mathbf{y} \leq \| \mathbf{x} \| \, \| \mathbf{y} \| \;\;\Rightarrow\;\; -1 \leq \frac{\mathbf{x} \cdot \boldsymbol{y}}{\| \mathbf{x} \| \, \| \boldsymbol{y} \|} \leq 1.$$

The final inequality says that the quantity $\mathbf{x}\cdot\mathbf{y}/(\|\mathbf{x}\|\,\|\mathbf{y}\|)$ lies between the minimum value –1 and the maximum value 1 of the cosine function. Since the range of the cosine function is the entire interval [–1, 1], there is then a unique real number θ between 0 and π for which $\cos\theta = \mathbf{x}\cdot\mathbf{y}/\|\mathbf{x}\|\,\|\mathbf{y}|$. This generalization of (6) leads to the following definition.

2.7. Definition. If **x** and **y** are nonzero vectors in $\mathbf{R}^n$, then the *angle between them* is

$$\theta = \text{ArcCos}\,\frac{\mathbf{x}\cdot\boldsymbol{y}}{\|\mathbf{x}\|\,\|\boldsymbol{y}\|} \tag{8}$$

The relation of the dot product to the magnitude of a vector makes the following extension of the triangle inequality for real numbers easy to derive.

2.10. Theorem. For any two vectors **x** and **y** in $\mathbf{R}^n$, $\|\mathbf{x}+\mathbf{y}\| \le \|\mathbf{x}\| + \|\mathbf{y}\|$.

Proof. Parts (b) and (e) of Theorem 2.4 give

$$\begin{aligned}\|\boldsymbol{x}+\boldsymbol{y}\|^2 &= (\mathbf{x}+\mathbf{y})\cdot(\mathbf{x}+\mathbf{y}) = (\mathbf{x}+\mathbf{y})\cdot\mathbf{x} + (\mathbf{x}+\mathbf{y})\cdot\mathbf{y}\\ &= \mathbf{x}\cdot\mathbf{x} + \mathbf{y}\cdot\mathbf{x} + \mathbf{x}\cdot\mathbf{y} + \mathbf{y}\cdot\mathbf{y}\\ &= \mathbf{x}\cdot\mathbf{x} + 2\mathbf{x}\cdot\mathbf{y} + \mathbf{y}\cdot\mathbf{y} \quad \text{by Theorem 2.4(a).}\\ &\le \|\boldsymbol{x}\|^2 + 2|\mathbf{x}\cdot\mathbf{y}| + \|\boldsymbol{y}\|^2 \le \|\boldsymbol{x}\|^2 + 2\|\mathbf{x}\|\,\|\mathbf{y}\| + \|\boldsymbol{y}\|^2\\ &\qquad \text{by the Cauchy-Schwarz inequality}\\ &\le (\|\boldsymbol{x}\| + \|\boldsymbol{y}\|)^2.\end{aligned}$$

Taking positive square roots of the first and last terms finishes the proof. QED

The Pythagorean theorem is an immediate consequence of Theorem 2.10.

2.11. Theorem. If **x** is perpendicular to **y**, then $\|\boldsymbol{x}+\boldsymbol{y}\|^2 = \|\boldsymbol{x}\|^2 + \|\boldsymbol{y}\|^2$.

Proof. Note that **x** perpendicular to **y** is equivalent to $\mathbf{x}\cdot\mathbf{y} = 0$. Thus in the proof of Theorem 2.10, the terms $\mathbf{y}\cdot\mathbf{x}$ and $\mathbf{x}\cdot\mathbf{y}$ in the first line are both 0. That line then reduces to the assertion that $\|\boldsymbol{x}+\boldsymbol{y}\|^2 = \|\boldsymbol{x}\|^2 + \|\boldsymbol{y}\|^2$. QED

The concept of parallel vectors in the plane extends directly to nonzero vectors of any dimension.

3.7. Definition. Two nonzero vectors **v** and **w** are ***parallel*** if their direction vectors $\mathbf{u_v}$ and $\mathbf{u_w}$ coincide or differ only in sign (that is, by a factor of –1).

There is a simple criterion for deciding whether two vectors are parallel:

v is parallel to **w** if and only if **w** = *a***v** for some nonzero scalar *a*.

Projection.Projecting a nonzero vector **x** onto a vector **v** is an important tool in physics, mathematics and statistics. Suppose that, as in the figure, **p** is the perpendicular projection of **x** onto **v**.



Then **p** and **v** have the same direction vector, namely,

$$\mathbf{u} = \frac{1}{\|\boldsymbol{v}\|}\mathbf{v} = \frac{1}{\|\boldsymbol{p}\|}\mathbf{p}.$$

From the figure it is also clear that

$$\|\mathbf{p}\| = \|\mathbf{x}\| \cos\theta = \|\mathbf{x}\| \frac{\mathbf{x}\cdot\mathbf{v}}{\|\boldsymbol{x}\|\,\|\boldsymbol{v}\|} = \frac{\mathbf{x}\cdot\mathbf{v}}{\|\boldsymbol{v}\|},$$

that is,

$$\mathbf{p} = \|\mathbf{p}\|\,\mathbf{u} = \frac{\mathbf{x}\cdot\mathbf{v}}{\|\boldsymbol{v}\|}\mathbf{u}.$$

This leads to the following definition.

3.8. Definition. If **x** and **v** are nonzero vectors in $\mathbf{R}^n$, then the ***coordinate of* x *in the direction of* v (**or ***scalar projection of* x *onto* v**) is the scalar

(9) $$x_v = \frac{\mathbf{x}\cdot\mathbf{v}}{\|\boldsymbol{v}\|}.$$

The ***(vector) projection of* x *onto vector* v** (or, ***the component of* x *in the direction of* v**) is

(10) $$\mathbf{proj}_{v}(\mathbf{x}) = \mathbf{p_v}(\mathbf{x}) = \frac{\mathbf{x}\cdot\mathbf{v}}{\|\boldsymbol{v}\|}\,\frac{1}{\|\mathbf{v}\|}\,\mathbf{v} = \frac{\mathbf{x}\cdot\mathbf{v}}{\boldsymbol{v}\cdot\boldsymbol{v}}\,\mathbf{v}.$$

Example. Find the projection of $\mathbf{x} = 3\mathbf{i} + \mathbf{j}$ onto $\mathbf{v} = -2\mathbf{i} + \mathbf{j}$.

Solution. The following *Mathematica* program prints the given vectors, computes and prints the projection of **x** onto **v**, and also displays a figure showing the two vectors and that projection. Execute it to view the coordinates of a picture of **x**, **v** and $\mathbf{proj}_{v}(\mathbf{x})$.

```
x = {3, 1}
v = {-2, 1}
p = x.v v / v.v
yht = Max[x[[2]], v[[2]]] / Max[v.v, x.x];
Needs["Graphics`Arrow`"]
Show[Graphics[{Arrow[{0, 0}, v],
   Text[FontForm["v", {"Times-Bold", 12}], {.8 v[[1]], .8 v[[2]] + yht}],
   {Dashing[{0.03, 0.03}], Line[{x, p}]},
   RGBColor[1, 0, 1], Arrow[{0, 0}, x], Arrow[{0, 0}, p],
   Text[FontForm["p", {"Times-Bold", 12}], {.8 p[[1]], .8 p[[2]] + yht}],
   Text[FontForm["x", {"Times-Bold", 12}],
    {.8 x[[1]], .8 x[[2]] + yht}]}, AspectRatio → Automatic,
  Axes → True, AxesLabel → {"x", "y"}]]
```

```
{3, 1}

{-2, 1}

{2, -1}
```



As usual, while *Mathematica* can do a 3-dimensional computation of projections just as quickly as a 2-dimensional one, generating a 3-dimensional diagram requires a more complicated set of commands.

Example. Find the projection of **x** = –2**i** + 4**j** + 10**k** onto **v** = 12**i** + 9**k**.

Solution. From (10),

$$\mathbf{p} = \mathbf{proj}_{v}(\mathbf{x}) = \frac{\mathbf{x}\cdot\mathbf{v}}{\mathbf{v}\cdot\mathbf{v}}\,\mathbf{v}$$

$$= \frac{(-2\cdot 12) + 4\cdot 0 + 10\cdot 9)}{(12)\,(12) + 9\cdot 9}\,\mathbf{v} = \frac{66}{225}\,(12, 0, 9)$$

$$= \frac{22}{75}\,(12, 0, 9) = (\frac{88}{25}, 0, \frac{66}{25}) = (3.52, 2.64).$$

The following *Mathematica* program displays the vectors **x** and **v**, checks the above computation, and plots the vectors **x**, **v**, and **p**.

NOTE: Some adjustment of these commands will help to display arrowheads and labels more reasonably for vectors of lengths different from those in this example. In particular, note the absolute units in the labeling `Text` commands. These were added after an initial plot that contained poor placement of the labels. Adjust them for other plots.

```
x = {-2, 4, 10}
v = {12, 0, 9}
p = (x.v)/(v.v) v
r = Max[x[[3]], v[[3]]]/Max[v.v, x.x];
xwin = 6;
yext = 6;
zlen = 5;
coordaxes = Graphics3D[
  {{RGBColor[0, 1, 0], Line[{{-xwin, 0, 0}, {xwin, 0, 0}}],
    Text["x", {xwin + xwin/5, 0, 0}]},
   {RGBColor[0, 1, 0], Line[{{0, -yext, 0}, {0, yext, 0}}],
    Text["y", {0, yext + yext/5, 0}]},
   {RGBColor[0, 1, 0], Line[{{0, 0, -zlen}, {0, 0, zlen}}],
    Text["z", {0, 0, zlen + zlen/5}]}}];
vplot = Graphics3D[{{RGBColor[0, 0, 1],
```

```
            Line[{{0, 0, 0}, x}],
  Polygon[{x, {.9 x[[1]] - yext/16, .9 x[[2]] - xwin/16, .9 x[[3]] - 3 r},
    {.9 x[[1]] + yext/16, .9 x[[2]] + xwin/16, .9 x[[3]] - 3 r}}],
  Text[FontForm["x", {"Times-Bold", 12}], {.5 x[[1]] + 1,
    .5 x[[2]], .5 x[[3]] + r}], {Dashing[{0.03}], Line[{x, p}]}},
 {RGBColor[1, 0, 1], Line[{{0, 0, 0}, v}],
  Polygon[{v, {.9 v[[1]] - yext/16, .9 v[[2]] + xwin/16, .9 v[[3]] - r},
    {.9 v[[1]] + yext/16, .9 v[[2]] - xwin/16, .9 v[[3]] - r}}],
  Text[FontForm["v", {"Times-Bold", 12}],
   {.8 v[[1]], .8 v[[2]], .8 v[[3]] + 1}]},
 {RGBColor[1, 0, 1], Line[{{0, 0, 0}, p}],
  Polygon[{p, {.7 p[[1]] - yext/12, .7 p[[2]] + xwin/12, .7 p[[3]] - r},
    {.7 p[[1]] + yext/12, .7 p[[2]] - xwin/12, .7 p[[3]] - r}}],
  Text[FontForm["p", {"Times-Bold", 12}],
   {.5 p[[1]], .5 p[[2]], .5 p[[3]] + 1}]}}];
Show[vplot, coordaxes, Axes → True, Lighting → False]
```